
CHAPTER 14

Everyday functions

thiS Chapter describes some of the more useful functions Microsoft Excel has to offer. To 
keep this book from threatening the structural integrity of your bookshelf, we had to 
make some hard choices about which functions to highlight. Therefore, this chapter 

(along with Chapters 15, 16, and 17) by no means represents a comprehensive reference. 
For complete information about all the built-in functions, you can use a number of on-
screen tools, which are covered in “Using the built-in function reference in Excel” in Chapter 
13, “Using functions.” Appendix C, “Function reference,” lists every function available in 
Excel, along with the basic information you need to put each one to use, and it includes 
cross-references to any information available in this and other chapters.

For more information, see Chapter 15, “Formatting and calculating date and time,” Chapter 
16, “Functions for financial analysis,” and Chapter 17, “Functions for analyzing statistics.”

Understanding mathematical functions
Most of the work you do in Excel probably involves at least a few mathematical functions. 
The most popular among these is the SUM function, but Excel is capable of calculating just 
about anything. In the next sections, we discuss some of the most used (and most useful) 
mathematical functions in Excel.

Using the SUM function
The SUM function totals a series of numbers. It takes the form =SUM(number1, number2, …). 
The number arguments are a series of as many as 30 entries that can be numbers, formulas, 
ranges, or cell references that result in numbers. SUM ignores arguments that refer to text 
values, logical values, or blank cells.

The Sum button

Because SUM is such a commonly used function, Excel provides the Sum button on the 
Home tab on the ribbon, as well as the AutoSum button on the Formulas tab. In addition to 

SUM, these buttons include a menu of other commonly used functions. If you select a cell 
and click the Sum button, Excel creates a SUM formula and guesses which cells you want to 
total. To enter SUM formulas in a range of cells, select the cells before clicking Sum.

The SUMIF, SUMIFS, and COUNTIF functions

The SUMIF function is similar to SUM, but it first tests each cell using a specified condi-
tional test before adding it to the total. This function takes the arguments (range, criteria, 
sum_range). The range argument specifies the range you want to test, the criteria argument 
specifies the conditional test to be performed on each cell in the range, and the sum_range 
argument specifies the cells to be totaled. For example, if you have a worksheet with a col-
umn of month names defined using the range name Months and an adjacent column of 
numbers named Sales, use the formula =SUMIF(Months, "June", Sales) to return the value in 
the Sales cell that is adjacent to the label June. Alternatively, you can use a conditional test 
formula such as =SUMIF(Sales, ">=999", Sales) to return the total of all sales figures that are 
more than $999.

The SUMIFS function does similar work to that of the SUMIF function, except you can 
specify up to 127 different ranges to sum, each with their own criteria. Note that in this 
function, the sum_range argument is in the first position instead of the third position: 
(sum_range, criteria_range1, criteria1, criteria_range2, criteria2, …). The sum range and each 
criteria range must all be the same size and shape. Using a similar example to the one 
we used for the SUMIF function, suppose we also created defined names for cell ranges 
Months, Totals, Product1, Product2, and so on. The formula =SUMIFS(Totals,  Product3, 
"<=124", Months, "June") returns the total sales for the month of June when sales of 
 Product3 were less than or equal to $124.

Similarly, COUNTIF counts the cells that match specified criteria and takes the arguments 
(range, criteria). Using the same example, you can find the number of months in which total 
sales fell to less than $600 by using a conditional test, as in the formula =COUNTIF(Totals, 
"<600").

For more information about conditional tests, see “Creating conditional tests” in Chapter 12, 
“Building formulas .” For more about using range names, see “Naming cells and cell ranges” in 
Chapter 12 .

Using selected mathematical functions
Excel has over 60 built-in math and trigonometry functions; the following sections brush 
only the surface, covering a few of the more useful or misunderstood functions. You can 
access them directly by clicking the Math & Trig button on the Formulas tab on the ribbon.

The AGGREGATE function

The AGGREGATE function, new in Excel 2013, returns an aggregate of a range or array by 
applying one of 19 functions, and it offers the option to ignore hidden rows, error values, 
or both. This function is designed for use on columns of data; it is not intended for use with 
rows, or ranges of data that are arranged horizontally.

When used with lists and ranges, the AGGREGATE function takes the arguments 
(function_num, options, ref1, ref2, …), where function_num is a number from 1 to 19 indicat-
ing the function you want to apply to the selected range, as shown in Table 14-1; options 
is an optional number from 1 to 7 indicating specific instructions for ignoring particular 
values, as shown in Table 14-2; and ref1, ref2, and the like are the cells or ranges you want 
to aggregate.

TABLE 14-1 AGGREGATE function numbers

Argument Description

1 AVERAGE
2 COUNT 
3 COUNTA
4 MAX
5 MIN
6 PRODUCT
7 STDEV
8 STDEV.P
9 SUM
10 VAR.S
11 VAR.P
12 MEDIAN
13 MODE.SNGL
14 LARGE  (ref2 required)
15 SMALL  (ref2 required)
16 PERCENTILE.INC  (ref2 required)
17 QUARTILE.INC  (ref2 required)
18 PERCENTILE.EXC  (ref2 required)
19 QUARTILE.EXC  (ref2 required)

A ref2 argument is required for functions 14 through 19, as indicated in Table 14-1. When 
used with arrays, the AGGREGATE function takes the arguments (function_num, options, 

array, k), where array is the array of values you want to aggregate and, depending on the 
function, k is a number indicating the percentile, quartile, or kth largest or kth smallest 
value you want to find.

TABLE 14-2 AGGREGATE options

Argument Description

0 or omitted Ignore nested SUBTOTAL and AGGREGATE functions
1 Ignore hidden rows, nested SUBTOTAL and AGGREGATE functions
2 Ignore error values, nested SUBTOTAL and AGGREGATE functions
3 Ignore hidden rows, error values, nested SUBTOTAL and AGGREGATE 

functions
4 Ignore nothing
5 Ignore hidden rows
6 Ignore error values
7 Ignore hidden rows and error values

The PRODUCT and SUMPRODUCT functions

The PRODUCT function multiplies all its arguments and can take as many as 255 arguments 
that are text or logical values; the function ignores blank cells.

You can use the SUMPRODUCT function to multiply the value in each cell in one range by 
the corresponding cell in another range of equal size and then add the results. You can 
include up to 255 arrays as arguments, but each array must have the same dimensions. 
(Non-numeric entries are treated as zero.) For example, the following formulas are essen-
tially the same:

=SUMPRODUCT(A1:A4, B1:B4)

{=SUM(A1:A4*B1:B4)}

The only difference between them is that you must enter the SUM formula as an array by 
pressing Ctrl+Shift+Enter.

For more information about arrays, see “Using arrays” in Chapter 12 .

The MOD function

The MOD function returns the remainder of a division operation (modulus). It takes the 
arguments (number, divisor). The result of the MOD function is the remainder produced 
when number is divided by divisor. For example, the function =MOD(9, 4) returns 1, the 
remainder that results from dividing 9 by 4.

A MOD example
Here’s a practical use of the MOD function that you can ponder:

1. Select a range of cells such as B5:F16, click Conditional Formatting on the Home 
tab on the ribbon, and then click New Rule .

2. Select the Use A Formula To Determine Which Cells To Format option in the Select 
A Rule Type list .

3. In the text box, type the formula =MOD(ROW(), 2)=0 .

4. Click the Format button, and select a color on the Fill tab to create a format that 
applies the selected color to every other row . Note that if you select a single cell 
in an odd-numbered row before creating this formatting formula, nothing seems 
to happen, but if you copy or apply the format to other rows, you’ll see the result . 
Click OK to accept the format, and then click OK to apply the formatting .

We clicked the Conditional Formatting button and clicked Manage Rules to display the 
dialog box shown in the preceding figure. The MOD formula identifies the current row 
number using the ROW function, divides it by 2, and if there is a remainder (indicating 
an odd-numbered row), it returns FALSE because the formula also contains the con-
ditional test =0 . If MOD returns anything but 0 as a remainder, the condition returns 
FALSE . Therefore, Excel applies formatting only when the formula returns TRUE (in 
even-numbered rows) . For more information about conditional formatting, see “For-
matting conditionally” in Chapter 9, “Worksheet formatting techniques .” You can also 
achieve similar results (with additional functionality) by converting the cell range into 
a table and using the table formatting features . For more information, see Chapter 22, 
“Managing information in tables .”

The COMBIN function

The COMBIN function determines the number of possible combinations, or groups, that 
can be taken from a pool of items. It takes the arguments (number, number_chosen), where 
number is the total number of items in the pool and number_chosen is the number of 
items you want to group in each combination. For example, to determine how many dif-
ferent 11-player football teams you can create from a pool of 17 players, type the formula 
=COMBIN(17, 11). The result indicates that you could create 12,376 teams.

Try your luck
The COMBIN function can help you figure out just how slim a chance you have of 
drawing the elusive ace-high straight flush in a game of five-card stud. You express 
the number of 5-card combinations by using the formula =COMBIN(52, 5), result-
ing in 2,598,960 . That’s not too bad when you consider the odds of winning the lot-
tery . For a lotto game in which you select 6 numbers out of a total of 49, the formula 
=COMBIN(49, 6) reveals that there are 13,983,816 possibilities . You’d better keep your 
day job either way .

The RAND and RANDBETWEEN functions

The RAND function generates a random number between 0 and 1. It’s one of the few Excel 
functions that doesn’t take an argument, but you must still type a pair of parentheses after 
the function name. The result of a RAND function changes each time you recalculate your 
worksheet. This is called a volatile function. If you use automatic recalculation, the value of 
the RAND function changes each time you make a worksheet entry.

The RANDBETWEEN function provides more control than RAND. With RANDBETWEEN, 
you can specify a range of numbers within which to generate random integer values. 
The arguments (bottom, top) represent the smallest and largest integers that the func-
tion should use. The values for these arguments are inclusive. For example, the formula 
=RAND BETWEEN(123, 456) can return any integer from 123 up to and including 456.

Using the rounding functions
Excel includes several functions devoted to the seemingly narrow task of rounding numbers 
by a specified amount.

The ROUND, ROUNDDOWN, and ROUNDUP functions

The ROUND function rounds a value to a specified number of decimal places. Digits to the 
right of the decimal point that are less than 5 are rounded down, and digits greater than or 
equal to 5 are rounded up. It takes the arguments (number, num_digits). If num_digits is a 
positive number, then number is rounded to the specified number of decimal points; if num_
digits is negative, the function rounds to the left of the decimal point; if num_digits is 0, the 
function rounds to the nearest integer. For example, the formula =ROUND(123.4567, –2) 
returns 100, and the formula =ROUND(123.4567, 3) returns 123.457. The ROUNDDOWN 
and ROUNDUP functions take the same form as ROUND. As their names imply, they always 
round down or up, respectively.

CAUTION!
Don’t confuse the rounding functions with rounded number formats, such as the one 
applied when you click the Accounting Number Format button in the Number group 
on the Home tab. When you format the contents of a cell to a specified number of dec-
imal places, you change only the display of the number in the cell; you don’t change 
the cell’s value . When performing calculations, Excel always uses the underlying value, 
not the displayed value . Conversely, the rounding functions permanently change the 
underlying values .

The EVEN and ODD functions

The EVEN function rounds a number up to the nearest even integer. The ODD function 
rounds a number up to the nearest odd integer. Negative numbers are correspond-
ingly rounded down. For example, the formula =EVEN(22.4) returns 24, and the formula 
=ODD(–4) returns –5.

The FLOOR and CEILING functions

The FLOOR function rounds a number down to its nearest given multiple, and the CEILING 
function rounds a number up to its nearest given multiple. These functions take the argu-
ments (number, multiple). For example, the formula =FLOOR(23.4, 0.5) returns 23, and the 
formula =CEILING(5, 1.5) returns 6, the nearest multiple of 1.5. The FLOOR.PRECISE and 
CEILING.PRECISE functions round numbers down or up to the nearest integer or multiple 
of significance. Both take the arguments (number, significance). For example, the formula 
=FLOOR.PRECISE(23.4, 4) returns 20, which is the nearest integer below 23.4 that is a mul-
tiple of 4. Most of the time, you see no difference in results between the regular and precise 
versions of these functions, unless your arguments are negative numbers. The precise ver-
sions always round up, regardless of the number’s sign.

Using the flexible MROUND function
Suppose you want to round a number to a multiple of something other than 10—for 
example, rounding numbers to sixteenths so that when formatted as fractions they 
never appear with a denominator larger than 16 . The MROUND function rounds any 
number to a multiple you specify .

The function takes the form =MROUND(number, multiple) . For example, typing the 
formula =MROUND(A1, .0625) rounds the number displayed in cell A1 in increments 
of one-sixteenth . The function rounds up if the remainder after dividing number by 
multiple is at least half the value of multiple . If you want to apply this to an existing for-
mula, just wrap the MROUND formula around it by replacing A1 (in the example) with 
your formula .

The INT function

The INT function rounds numbers down to the nearest integer. For example, the formulas

=INT(100.01)

=INT(100.99999999)

both return the value 100, even though the number 100.99999999 is essentially equal to 
101. When a number is negative, INT also rounds that number down to the next integer. If 
each of the numbers in these examples were negative, the resulting value would be –101.

AVERAGE vs. AVG
Some other spreadsheet programs use the AVG statistical function to compute aver-
ages . In some previous versions of Excel, typing the formula =AVG(2, 4, 5, 8) results 
in a #NAME? error . Excel now accepts AVG, although when you type the function, an 
error dialog box appears, asking whether you want to change the function to AVERAGE . 
That’s still kind of rude, but it works . Presumably, one reason why Excel doesn’t just 
change AVG to AVERAGE for you is to remind you to start using the correct function 
name .

When you use this function, Excel ignores cells containing text, logical values, or empty 
cells, but it includes cells containing a zero value . You can also choose the AVERAGEA 
function, which operates in the same way as AVERAGE, except it includes text and logi-
cal values in the calculation .

The TRUNC function

The TRUNC function truncates everything to the right of the decimal point in a number, 
regardless of its sign. It takes the arguments (number, num_digits). If num_digits isn’t speci-
fied, it’s set to 0. Otherwise, TRUNC truncates everything after the specified number of dig-
its to the right of the decimal point. For example, the formula =TRUNC(13.978) returns the 
value 13; the formula =TRUNC(13.978, 1) returns the value 13.9.

Understanding text functions
Text functions in Excel are some of the most useful word-processing and data-management 
tools you’ll find anywhere—they perform tasks that word-processing programs can’t. These 
functions are conveniently listed for you when you click the Text button on the Formulas 
tab on the ribbon.

You can use the TRIM and CLEAN functions to remove extra spaces and nonprinting char-
acters, which is great for cleaning up imported data—a task that ranges from difficult to 
impossible using search and replace. The UPPER, LOWER, and PROPER functions change the 
case of words, sentences, and paragraphs with no retyping. You might find yourself copying 
text from other documents into Excel just so that you can apply these functions. After using 
text functions, select the cells containing the formulas, press Ctrl+C to copy, click the Paste 
button on the Home tab, and then click Paste Values to convert the formulas to their result-
ing (text) values. You can then copy the edited text into the original document.

In the following sections, we discuss the most useful Excel text functions.

Using selected text functions
Text functions convert numeric entries, as well as numeric text entries, into text strings so 
that you can manipulate the text strings themselves. Numeric text is a type of numeric 
entry that provides a few specific text characters in addition to numeric characters. For 
details, see “Using numeric text in formulas” in Chapter 12.

The TEXT function

The TEXT function converts a number into a text string with a specified format. Its argu-
ments are (value, format_text), where value represents any number, formula, or cell refer-
ence; and format_text is the format for displaying the resulting string. For example, the 
formula =TEXT(98/4, "0.00") returns the text string 24.50. You can use any Excel formatting 
symbol ($, #, 0, and so on) except the asterisk (*) to specify the format you want, but you 
can’t use the General format.

For information about formatting symbols and codes, see Table 9-1, “Custom format sym-
bols,” and Table 9-2, “Built-in custom format codes,” both in Chapter 9 .

The DOLLAR function

Like the TEXT function, the DOLLAR function converts a number into a string. DOLLAR, 
however, formats the resulting string as currency with the number of decimal places you 
specify. The arguments (number, decimals) specify a number or reference and the number 
of decimal places you want. For example, the formula =DOLLAR(45.899, 2) returns the text 
string $45.90. Notice that Excel rounds the number when necessary.

If you omit decimals, Excel uses two decimal places. If you add a comma after the first argu-
ment but omit the second argument, Excel uses zero decimal places. If you use a negative 
number for decimals, Excel rounds to the left of the decimal point.

The LEN function

The LEN function returns the number of characters in an entry. The single argument can 
be a number, a string enclosed in double quotation marks, or a reference to a cell. Trailing 
zeros are ignored. For example, the formula =LEN("Test") returns 4.

The LEN function returns the length of the displayed text or value, not the length 
of the underlying cell contents. For example, suppose cell A10 contains the formula 
=A1+A2+A3+A4+A5+A6+A7+A8 and its result is the value 25. The formula =LEN(A10) 
returns the value 2, which indicates the length of the resulting value, 25. The cell referenced 
as the argument of the LEN function can contain another string function. For example, if 
cell A1 contains the function =REPT("–*", 75), which enters the two-character hyphen and 
asterisk string 75 times in a cell, the formula =LEN(A1) returns the value 150.

The ASCII functions: CHAR and CODE

Every computer uses numeric codes to represent characters. The most prevalent system of 
numeric codes is ASCII, or American Standard Code for Information Interchange. ASCII uses 
a number from 0 to 127 (or to 255 in some systems) to represent each number, letter, and 
symbol.

The CHAR and CODE functions deal with these ASCII codes. The CHAR function returns the 
character that corresponds to an ASCII code number; the CODE function returns the ASCII 
code number for the first character of its argument. For example, the formula =CHAR(83) 
returns the text S. The formula =CODE("S") returns the ASCII code 83. If you type a literal 
character as the text argument, be sure to enclose the character in quotation marks; other-
wise, Excel returns the #NAME? error value.

Note
If you use certain ASCII symbols often, you can use the ASCII code number with the 
CHAR function to create a symbol without using the Symbol button on the Insert tab 
on the ribbon . For example, to create a registered trademark symbol (®), just type 
=CHAR(174) .

The cleanup functions: TRIM and CLEAN

Leading and trailing blank characters often prevent you from correctly sorting entries in a 
worksheet or a database. If you use string functions to manipulate text in your worksheet, 
extra spaces can prevent your formulas from working correctly. The TRIM function elimi-
nates leading, trailing, and extra blank characters from a string, leaving only single spaces 
between words.

The CLEAN function is similar to TRIM, except it operates only on nonprintable characters, 
such as tabs and program-specific codes. CLEAN is especially useful if you import data from 
another program or operating system, because the translation process often introduces 
nonprintable characters that appear as symbols or boxes. You can use CLEAN to remove 
these characters from the data.

The EXACT function

The EXACT function is a conditional function that determines whether two strings match 
exactly. The function ignores formatting, but it is case sensitive, so uppercase letters are 
considered different from lowercase letters. If both strings are identical, the function returns 
TRUE. Both arguments must be literal strings enclosed in quotation marks, references to 
cells that contain text, numeric values, or formulas that evaluate to numeric values. For 
example, if cell A5 and cell A6 on your worksheet both contain the text Totals, the formula 
=EXACT(A5, A6) returns TRUE.

For information about comparing strings, see “Creating conditional tests” in Chapter 12 .

The case functions: UPPER, LOWER, and PROPER

Three functions manipulate the case of characters in text strings. The UPPER and LOWER 
functions convert text strings to all uppercase or all lowercase letters. The PROPER func-
tion capitalizes the first letter in each word, capitalizes any other letters in the text string 
that do not follow another letter, and converts all other letters to lowercase. For example, 
if cell A1 contains the text mark Dodge, you can type the formula =UPPER(A1) to return 
MARK DODGE. Similarly, the formula =LOWER(A1) returns mark dodge, and =PROPER(A1) 
returns Mark Dodge.

Unexpected results can occur when the text contains an apostrophe, however. For example, 
if cell A1 contains the text it wasn’t bad, the PROPER function converts it to It Wasn’T Bad.

Using the substring text functions
The following functions locate and return portions of a text string or assemble larger 
strings from smaller ones: FIND, SEARCH, RIGHT, LEFT, MID, REPLACE, SUBSTITUTE, and 
CONCATENATE.

The FIND and SEARCH functions

You use the FIND and SEARCH functions to locate the position of a substring within a 
string. Both functions return the position in the string of the character you specify. (Excel 
counts blank spaces and punctuation marks as characters.) These two functions work the 
same way, except FIND is case sensitive and SEARCH allows wildcards. Both functions take 
the same arguments: (find_text, within_text, start_num). The optional start_num argument 
is helpful when within_text contains more than one occurrence of find_text. If you omit 
start_num, Excel reports the first match it locates. For example, to locate the p in the string 
A Night at the Opera, you type the formula =FIND("p", "A Night at the Opera"). The for-
mula returns 17, because p is the 17th character in the string.

If you’re not sure of the character sequence you’re searching for, you can use the SEARCH 
function and include wildcards in your find_text string. Suppose you used the names Smith 
and Smyth in your worksheet. To determine whether either name is in cell A1, type the for-
mula =SEARCH("Sm?th", A1). If cell A1 contains the text John Smith or John Smyth, the 
SEARCH function returns the value 6—the starting point of the string Sm?th.

If you’re not sure of the number of characters, use the * wildcard. For example, to find 
the position of Allan or Alan within the text (if any) stored in cell A1, type the formula 
=SEARCH("A*an", A1) (which would also return the word American).

Note
You might notice the presence of several text functions that look similar to others but 
end in “B,” such as FINDB, LEFTB, LENB, REPLACEB, and so on . These functions perform 
similar tasks to their “non-B” counterparts, but they are included specifically for use 
with double-byte character sets, such as Japanese, Chinese, and Korean .

The RIGHT and LEFT functions

The RIGHT function returns the rightmost series of characters from a specified string; the 
LEFT function returns the leftmost series of characters. These functions take the same argu-
ments: (text, num_chars). The num_chars argument indicates the number of characters to 
extract from the text argument.

Practical text manipulation
Excel excels at text manipulation . If you deal with a lot of mailing lists, for example, 
here is a trick you might find useful, and it serves as a good example as well. Sup-
pose you import a database of addresses in which the first and last names are stored 
together in the same cell . This example shows you a formula you can use to parse them 
into separate columns:

The formulas in columns C and D use the SEARCH function to locate the space charac-
ter . The function assigns it a number based on its position in the cell . Inside the RIGHT 
function shown in this figure, the number is subtracted from the total number of char-
acters in the cell, provided by the LEN function . This returns everything to the right 
of the space . The formula you can’t see in cell C2 doesn’t require the LEN function: 
=LEFT(A2,SEARCH(" ",A2)). You might want to use the TRIM function on column A first 
to be sure there are no extra spaces . For a related text-manipulation trick, see “Practical 
concatenation” in Chapter 12 . And for an even better trick that is new in Excel 2013, 
see “Automatic parsing and concatenation using Flash Fill” in Chapter 8, “Worksheet 
editing techniques .”

You’ll find the Text.xlsx file with the other examples on the companion website.

These functions count blank spaces in the text argument as characters; if text contains lead-
ing or trailing blank characters, you might want to use a TRIM function within the RIGHT or 
LEFT function to ensure the expected result. For example, suppose you type This is a test 
in cell A1 on your worksheet. The formula =RIGHT(A1, 4) returns the word test.

The MID function

You can use the MID function to extract a series of characters from a text string. This func-
tion takes the arguments (text, start_num, num_chars). For example, if cell A1 contains the 
text This Is A Long Text Entry, you can type the formula =MID(A1, 11, 9) to extract the 
characters Long Text from the entry in cell A1.

The REPLACE and SUBSTITUTE functions

The REPLACE and SUBSTITUTE functions substitute new text for old text. The REPLACE 
function replaces one string of characters with another string of characters and takes the 
arguments (old_text, start_num, num_chars, new_text). Suppose cell A1 contains the text 
Eric Miller, CEO. To replace the first four characters with the string Geof, type the formula 
=REPLACE(A1, 1, 4, "Geof"). The result is Geof Miller, CEO.

With the SUBSTITUTE function, you specify the text to replace. The function takes the argu-
ments (text, old_text, new_text, instance_num). Suppose cell A1 contains the text Mandy and 
you want to place it in cell A2 but change it to Randy. Type =SUBSTITUTE(A1, "M", "R") 
in cell A2.

The instance_num argument optionally replaces only the specified occurrence of 
old_text. For example, if cell A1 contains the text through the hoop, the 4 in the formula 
=SUBSTITUTE(A1, "h", "l", 4) tells Excel to substitute an l for the fourth h found in cell A1. If 
you don’t include instance_num, Excel changes all occurrences of old_text to new_text.

Note
You can create an array formula using the SUBSTITUTE function to count the number of 
occurrences of a text string in a range of cells . Use the formula =SUM(LEN(<range>)–
LEN(SUBSTITUTE(<range>, "text", "")))/LEN("text") to count the number of times text 
appears in <range> . Type the formula, and press Ctrl+Shift+Enter .

The CONCATENATE function

To assemble strings from as many as 255 smaller strings or references, the CONCATENATE 
function is the function equivalent of the & character. For example, if cell B4 contains the 
text Salt, the formula =CONCATENATE(B4, " of the Earth") returns Salt of the Earth. Note 
that we included a leading space character in the text argument; otherwise, the result 
would be Saltof the Earth. You could also specify the space character as a separate argu-
ment, as in the formula =CONCATENATE(B4, " ", "of the Earth").

TROUBLESHOOTING
Concatenated dates become serial numbers
If you try to concatenate the contents of a cell formatted as a date, the result is prob-
ably not what you expect . Because a date in Excel is only a serial number, what you 
usually see is a formatted representation of the date . But when you concatenate the 
contents of a date-formatted cell, you get the unformatted version of the date . To 
avoid this problem, use the TEXT function to convert the serial number to a recogniz-
able form . For example, suppose cell A1 contains the text Today’s Date is, and cell A2 
contains the function =NOW() and is formatted to display the date in dd/mm/yyyy 
format . Despite the cell’s formatting, the formula =CONCATENATE(A1, " ", A2) results in 
the value Today’s Date is 41573 (or whatever the current date serial number happens to 
be) . To remedy this problem, use the TEXT function as follows:

=CONCATENATE(A1, " ", TEXT(A2, "dd/mm/yyyy")) .

This version returns the value Today’s Date is 10/26/2013 (or whatever today’s date 
happens to be) . Note that the formula includes a space character (" ") as a separate 
argument between the two cell reference arguments .

Understanding logical functions
You use logical functions to test for specific conditions. These functions are often called 
logical operators in discussions of Boolean logic, which is named after George Boole, the 
British mathematician. You might have run across logical operators in set theory, which is 
used for teaching logical concepts in high school. You use logical operators to arrive at one 
of two conclusions: TRUE or FALSE. We discuss the most useful logical functions in the fol-
lowing sections. You can access the logical functions by clicking the Logical button on the 
Formulas tab on the ribbon.

Using selected logical functions
Excel has a rich set of logical functions. Most logical functions use conditional tests to 
determine whether a specified condition is TRUE or FALSE.

For more information about conditional tests, see “Creating conditional tests” in Chapter 12 .

The IF function

The IF function returns values according to supplied conditional tests. It takes the argu-
ments (logical_test, value_if_true, value_if_false). For example, the formula =IF(A6<22, 5, 
10) returns 5 if the value in cell A6 is less than 22; otherwise, it returns 10. You can nest 

other functions within an IF function. For example, the formula =IF(SUM(A1:A10)>0, 
SUM(A1:A10), 0) returns the sum of A1 through A10 if the sum is greater than 0; otherwise, 
it returns 0.

You can also use text arguments to return nothing instead of zero if the result is false. For 
example, the formula =IF(SUM(A1:A10)>0, SUM(A1:A10), " ") returns a null string (" ") if the 
conditional test is false. The logical_test argument can also consist of text. For example, the 
formula =IF(A1="Test", 100, 200) returns the value 100 if cell A1 contains the string Test; it 
returns 200 if A1 contains any other entry. The match between the two text entries must be 
exact except for case.

INSIDE OUT   Streamline formulas using the SUMIF function

If you find yourself frequently using the IF function to perform conditional tests on 
individual rows or columns and then use the SUM function to total the results, the 
SUMIF function might make your work a little easier. With SUMIF, you can add specific 
values in a range according to a criterion you supply . For example, you can type the 
formula =SUMIF(C12:C27, "Yes", A12:A27) to find the total of all numbers in A12:A27 
in which the cell in the same row in column C contains the word Yes . This performs all 
the calculations you need in one cell and eliminates having to create a column of IF 
formulas . For more information about SUMIF, see “The SUMIF, SUMIFS, and COUNTIF 
functions” later in this chapter .

The AND, OR, and NOT functions

These three functions help you develop compound conditional tests. They work with the 
logical operators =, >, <, >=, <=, and <>. The AND and OR functions can each have as 
many as 255 logical arguments. The NOT function takes only one argument. Arguments 
can be conditional tests, arrays, or references to cells that contain logical values.

Suppose you want Excel to return the text Pass only if a student has an average score 
greater than 75 and fewer than five unexcused absences. In Figure 14-1, we typed the for-
mula =IF(AND(G4<5,F4>75), "Pass", "Fail"). This fails the student in row 5 because of 
the five absences. If you use OR instead of AND in the formula shown in Figure 14-1, all 
students would pass.

Figure 14-1 You can create complex conditional tests using the AND function.

You’ll find the And Or Not.xlsx file with the other examples on the companion website.

The OR function returns the logical value TRUE if any one of the conditional tests is true; 
the AND function returns the logical value TRUE only if all the conditional tests are true.

Because the NOT function negates a condition, you usually use it with other functions. NOT 
instructs Excel to return the logical value TRUE if the argument is false or the logical value 
FALSE if the argument is true. For example, the formula =IF(NOT(A1=2), "Go", " ") tells Excel 
to return the text Go if the value of cell A1 is anything but 2.

Nested IF functions

Sometimes you can’t resolve a logical problem using only logical operators and the AND, 
OR, and NOT functions. In these cases, you can nest IF functions to create a hierarchy of 
tests. For example, the formula =IF(A1=100, "Always", IF(AND(A1>=80, A1<100), "Usually", 
IF(AND(A1>=60, A1<80), "Sometimes", "Who cares?"))) states, in plain language, the fol-
lowing: If the value is 100, return Always; if the value is from 80 through 99, return Usually; 
if the value is from 60 through 79, return Sometimes; or if none of these conditions is true, 
return Who cares?. You can create formulas containing up to 64 levels of nested functions.

Other uses for conditional functions

You can use all the conditional functions described in this section as stand-alone formulas. 
Although you usually use functions such as AND, OR, NOT, ISERROR, ISNA, and ISREF within 
an IF function, you can also use formulas, such as =AND(A1>A2, A2<A3), to perform simple 
conditional tests. This formula returns the logical value TRUE if the value in A1 is greater 
than the value in A2 and the value in A2 is less than the value in A3. You might use this type 
of formula to assign TRUE and FALSE values to a range of numeric database cells and then 
use the TRUE and FALSE conditions as selection criteria for printing a specialized report.

Understanding information functions
The information functions can be considered the internal monitoring system in Excel. 
Although they perform no specific calculations, you can use them to find out about ele-
ments of the Excel interface and then use that information elsewhere. We discuss the 
most useful of these functions in the following sections. You can find these functions by 
clicking the More Functions button on the Formulas tab on the ribbon and then clicking 
Information.

Using selected information functions
With information functions, you can gather information about the contents of cells, their 
formatting, and the computing environment and also perform conditional tests for the 
presence of specific types of values.

The TYPE and ERROR .TYPE functions

The TYPE function determines whether a cell contains text, a number, a logical value, an 
array, or an error value. The result is a code for the type of entry in the referenced cell: 1 for 
a number (or a blank cell), 2 for text, 4 for a logical value (TRUE or FALSE), 16 for an error 
value, and 64 for an array. For example, if cell A1 contains the number 100, the formula 
=TYPE(A1) returns 1. If A1 contains the text Microsoft Excel, the formula returns 2.

Like the TYPE function, the ERROR.TYPE function detects the contents of a cell, except it 
detects different types of error values. The result is a code for the type of error value in the 
referenced cell: 1 for #NULL!, 2 for #DIV/0!, 3 for #VALUE!, 4 for #REF!, 5 for #NAME!, 6 
for #NUM!, and 7 for #N/A. Any other value in the referenced cell returns the error value 
#N/A. For example, if cell A1 contains a formula that displays the error value #NAME!, the 
formula =ERROR.TYPE(A1) returns 5. If A1 contains the text Microsoft Excel, the formula 
returns #N/A.

The COUNTBLANK function

The COUNTBLANK function counts the number of empty cells in the specified range, 
which is its only argument. This function is tricky because formulas that evaluate to null 
text strings, such as =" ", or to zero might seem empty, but they aren’t and therefore aren’t 
counted.

Using the IS information functions
You can use the ISBLANK, ISERR, ISERROR, ISEVEN, ISFORMULA, ISLOGICAL, ISNA, ISNON-
TEXT, ISNUMBER, ISODD, ISREF, and ISTEXT functions to determine whether a referenced 
cell or range contains the corresponding type of value.

All the IS information functions take a single argument. For example, the ISBLANK function 
takes the form =ISBLANK(value). The value argument is a reference to a cell. If value refers 
to a blank cell, the function returns the logical value TRUE; otherwise, it returns FALSE. Note 
that when you type numeric values as text, such as ="21", the IS functions, unlike other 
functions, do not recognize them as numbers. Therefore, the formula =ISNUMBER("21") 
returns FALSE.

TROUBLESHOOTING
My IS function returns unexpected results
Although you can use a cell range (rather than a single cell) as the argument to any IS 
function, the result might not be what you expect . For example, you might think the 
ISBLANK function returns TRUE if the referenced range is empty or FALSE if the range 
contains any values . Instead, the function’s behavior depends on where the range is in 
relation to the cell containing the formula . If the argument refers to a range that inter-
sects the row or column containing the formula, ISBLANK uses implicit intersection to 
arrive at the result . In other words, the function looks at only one cell in the referenced 
range and only if it happens to be in the same row or column as the cell containing the 
function . The function ignores the rest of the range . If the range shares neither a row 
nor a column with the formula, the result is always FALSE . For more about intersection, 
see “Getting explicit about intersections” in Chapter 12 .

An ISERR example
You can use ISERR to avoid getting error values as formula results . For example, sup-
pose you want to call attention to cells containing a particular character string, such as 
12A, resulting in the word Yes appearing in the cell containing the formula . If the string 
isn’t found, you want the cell to remain empty . You can use the IF and FIND functions 
to perform this task, but if the value isn’t found, you get a #VALUE! error rather than a 
blank cell .

To solve this problem, add an ISERR function to the formula . The FIND function returns 
the position at which a substring is found within a larger string . If the substring 
isn’t there, FIND returns #VALUE! . The solution is to add an ISERR function, such as 
=IF(ISERR(FIND("12A", A1)), " ", "Yes") . Because you’re not interested in the error, which 
is simply a by-product of the calculation, this traps the error, leaving only the results in 
which you are interested .

Understanding lookup and reference functions
Lookup and reference functions help you use your own worksheet tables as sources of 
information to be used elsewhere in formulas. You can use three primary functions to look 
up information stored in a list or a table or to manipulate references: LOOKUP, VLOOKUP, 
and HLOOKUP. Some powerful lookup and reference functions in addition to these three 
are available; we describe many of them in the following sections. You can find a list of all 
these functions by clicking the Lookup & Reference button on the Formulas tab on the 
ribbon.

Using selected lookup and reference functions
VLOOKUP and HLOOKUP are nearly identical functions that look up information stored in 
tables you have constructed. VLOOKUP and HLOOKUP operate in either vertical or horizon-
tal orientation (respectively), but LOOKUP works either way.

When you look up information in a table, you usually use a row index and a column index 
to locate a particular cell. Excel derives the first index by finding the largest value in the 
first column or row that is less than or equal to a lookup value you supply, and then it uses 
a row number or column number argument as the other index. You need to make sure 
the table is sorted by the row or column containing the lookup values first, or you will get 
unexpected results.

You can create powerful lookup formulas by using conditional tests . For more information, 
see “Creating conditional tests” in Chapter 12 .

These functions take the following forms:

=VLOOKUP(lookup_value, table_array, col_index_num, range_lookup)

=HLOOKUP(lookup_value, table_array, row_index_num, range_lookup)

Table 14-3 lists LOOKUP function arguments and their descriptions. The LOOKUP function 
takes two forms. The first is called the vector form, and the second is called the array form:

=LOOKUP(lookup_value, lookup_vector, result_vector)

=LOOKUP(lookup_value, array)

The difference between the lookup functions is the type of table each function uses: 
VLOOKUP works only with vertical tables (tables arranged in columns); HLOOKUP works 
only with horizontal tables (tables arranged in rows). You can use the array form of LOOKUP 
with either horizontal tables or vertical tables, and you can use the vector form with single 
rows or columns of data.

TABLE 14-3 LOOKUP function arguments

Argument Description

lookup_value The value, cell reference, or text (enclosed in quotation marks) you 
want to find in a table or a range.

table_array A cell range or name that defines the table in which to look.
row_index_num  
col_index_num

The row or column number in the table from which to select the 
result, counted relative to the table (not according to the actual row 
and column numbers).

range_lookup A logical value that determines whether the function matches 
lookup_value exactly or approximately. Type FALSE to match lookup_
value exactly. The default is TRUE, which finds the closest match.

lookup_vector A one-row or one-column range that contains numbers, text, or 
logical values.

result_vector A one-row or one-column range that must be the same size as 
lookup_vector.

array A range containing numbers, text, or logical values to compare with 
lookup_value.

The array form of LOOKUP determines whether to search horizontally or vertically based on 
the shape of the table defined in the array argument. If the table has more columns than 
rows, LOOKUP searches the first row for lookup_value; if the table has more rows than col-
umns, LOOKUP searches the first column for lookup_value. LOOKUP always returns the last 
value in the row or column containing the lookup_value argument, or you can specify a row 
or column number using VLOOKUP or HLOOKUP.

The VLOOKUP and HLOOKUP functions

For the VLOOKUP and HLOOKUP functions, whether Excel considers a lookup table to be 
vertical or horizontal depends on where the comparison values (the first index) are located. 
If the values are in the leftmost column of the table, the table is vertical; if they are in the 
first row of the table, the table is horizontal. (In contrast, LOOKUP uses the shape of the 
table to determine whether to use the first row or column as the comparison values.) The 
comparison values can be numbers or text, but they must be sorted in ascending order. No 
comparison value should be used more than once in a table.

The index_num argument (sometimes called the offset) provides the second index and tells 
the lookup function which column or row of the table to look in for the function’s result. 
The first column or row in the table has an index number of 1; therefore, the index_num 
argument must be greater than or equal to 1 and must never be greater than the number 
of rows or columns in the table. For example, if a vertical table is three columns wide, the 
index number can’t be greater than 3. If any value does not meet these rules, the function 
returns an error value.

You can use the VLOOKUP function to retrieve information from the table in Figure 14-2.

Figure 14-2 VLOOKUP returns a value in the same row as the lookup value.

You’ll find the Lookup.xlsx file with the other examples on the companion website.

Remember that these lookup functions usually search for the greatest comparison value 
that is less than or equal to the lookup value, not for an exact match between the com-
parison values and the lookup value. If all the comparison values in the first row or column 
of the table range are greater than the lookup value, the function returns the #N/A error 
value. If all the comparison values are less than the lookup value, however, the function 
returns the value that corresponds to the last (largest) comparison value in the table, which 
might not be what you want. If you require an exact match, type FALSE as the range_
lookup argument.

The worksheet in Figure 14-3 shows an example of a horizontal lookup table using the 
HLOOKUP function.

Figure 14-3 HLOOKUP returns a value in the same column as the lookup value.

The LOOKUP function

The LOOKUP function is similar to VLOOKUP and HLOOKUP, follows the same rules, and 
is available in the same two forms, vector and array, whose arguments are described in 
Table 14-3.

Like HLOOKUP and VLOOKUP, the vector form of LOOKUP searches for the largest com-
parison value that isn’t greater than the lookup value. It then selects the result from the 
corresponding position in the specified result range. The lookup_vector and result_vector 
arguments are often adjacent ranges, but they don’t have to be when you use LOOKUP. 
They can be in separate areas of the worksheet, and one range can be horizontal and the 
other vertical. The only requirement is that they must have the same number of elements.

For example, consider the worksheet in Figure 14-4, where the ranges are not parallel. 
Both the lookup_vector argument, A1:A5, and the result_vector argument, D6:H6, have five 
 elements. The lookup_value argument, 3, matches the entry in the third cell of the lookup_
vector argument, making the result of the formula the entry in the third cell of the result 
range: 300.

The array form of LOOKUP is similar to VLOOKUP and HLOOKUP, but it works with either a 
horizontal table or a vertical table, using the dimensions of the table to figure out the loca-
tion of the comparison values. If the table is taller than it is wide or the table is square, the 
function treats it as a vertical table and assumes that the comparison values are in the left-
most column. If the table is wider than it is tall, the function views the table as horizontal 
and assumes that the comparison values are in the first row of the table. The result is always 
in the last row or column of the specified table; you can’t specify column or row numbers.

Figure 14-4 The vector form of the LOOKUP function can retrieve information from a 
nonparallel cell range.

Because HLOOKUP and VLOOKUP are more predictable and controllable, you’ll generally 
find using them preferable to using LOOKUP.

The ADDRESS function

The ADDRESS function provides a handy way to build a cell reference by using numbers 
typed into the formula or using values in referenced cells. It takes the arguments (row_num, 
column_num, abs_num, a1, sheet_text). For example, the formula =ADDRESS(1, 1, 1, TRUE, 
"Data Sheet") results in the reference 'Data Sheet'v!$A$1.

The CHOOSE function

You use the CHOOSE function to retrieve an item from a list of values. The function takes 
the arguments (index_num, value 1, value 2, …) and can include up to 254 values. The 
index_num argument is the position in the list you want to return; it must be positive and 
can’t exceed the number of elements in the list. The function returns the value of the ele-
ment in the list that occupies the position indicated by index_num. For example, the for-
mula =CHOOSE(2, 6, 1, 8, 9, 3) returns the value 1, because 1 is the second item in the list. 
(The index_num value isn’t counted as part of the list.) You can use individual cell references 
for the list, but you can’t specify ranges. You might be tempted to create a formula such as 
=CHOOSE(A10, C1:C5) to take the place of the longer function in the preceding example. If 
you do, however, the result is a #VALUE! error value.

The MATCH function

The MATCH function is closely related to the CHOOSE function. However, where CHOOSE 
returns the item that occupies the position in a list specified by the index_num argument, 
MATCH returns the position of the item in the list that most closely matches a lookup value.

Note
You can create powerful lookup formulas by using the MATCH and INDEX functions . 
See “Using lookup functions” in Chapter 12 .

This function takes the arguments (lookup_value, lookup_array, match_type), where lookup_
value and the items in the lookup_array can be numeric values or text strings, and match_
type defines the rules for the search, as shown in Table 14-4.

When you use MATCH to locate text strings, you should specify a match_type argument of 
0 (an exact match). You can then use the wildcards * and ? in the lookup_value argument.

TABLE 14-4 MATCH function arguments

match_type Description

1 (or omitted) Finds the largest value in the specified range (which must be sorted in 
ascending order) that is less than or equal to lookup_value. If no items in 
the range meet these criteria, the function returns #N/A.

0 Finds the first value in the specified range (no sorting necessary) that 
is equal to lookup_value. If no items in the range match, the function 
returns #N/A.

–1 Finds the smallest value in the specified range (which must be sorted in 
descending order) that is greater than or equal to lookup_value. If no 
items in the range meet these criteria, the function returns #N/A.

The INDEX function

The INDEX function has two forms: an array form, which returns a value, and a reference 
form, which returns a cell reference. The forms of these functions are as follows:

=INDEX(array, row_num, column_num)

=INDEX(reference, row_num, column_num, area_num)

The array form works only with an array argument; it returns the value of the result, not 
the cell reference. The result is the value at the position in array indicated by row_num and 
 column_num. For example, the formula

=INDEX({10,20,30;40,50,60} , 1, 2)

returns the value 20, because 20 is the value in the cell in the second column and first row 
of the array.

Note
Each form of the INDEX function offers an advantageous feature . Using the reference 
form of the function, you can use multiple, nonadjacent areas of the worksheet as the 
reference lookup range . Using the array form of the function, you can get a range of 
cells, rather than a single cell, as a result .

The reference form returns a cell address instead of a value and is useful when you want 
to perform operations on a cell (such as changing the cell’s width) rather than on its value. 
This function can be confusing, however, because if an INDEX function is nested in another 
function, that function can use the value in the cell whose address is returned by INDEX. 
Furthermore, the reference form of INDEX doesn’t display its result as an address; it displays 

the value or values at that cell address. Remember that the result is an address, even if it 
doesn’t look like one.

Here are a few guidelines to keep in mind when using the INDEX function:

 ● If you type 0 as the row_num or column_num argument, INDEX returns a reference 
for the entire row or column, respectively.

 ● The reference argument can be one or more ranges, which are called areas. Each area 
must be rectangular and can contain numbers, text, or formulas. If the areas are not 
adjacent, you must enclose the reference argument in parentheses.

 ● You need the area_num argument only if you include more than one area in refer-
ence. The area_num argument identifies the area to which the row_num and column_
num arguments will be applied. The first area specified in reference is designated area 
1, the second area 2, and so on.

Let’s consider some examples to see how all this works. Figure 14-5 shows an example of an 
INDEX function. The formula in cell A1 uses the row coordinate in cell A2 and the column 
coordinate in cell A3 to return the contents of the cell in the third row and second column 
of the specified range.

Figure 14-5 Use the INDEX function to retrieve the address or value in a cell where information 
is located.

The following example is a bit trickier: Using the same worksheet as in Figure 14-5, the 
 formula =INDEX(C3:E6, 0, 2) displays the #VALUE! error value because the row_num argu-
ment of 0 returns a reference to the entire column specified by the column_num argument 
of 2, or the range D3:D6. Excel can’t display a range as the result. However, try nesting this 
formula in another function as follows: =SUM(INDEX(C3:E6, 0, 2)). The result is 2600, the 
sum of the values in D3:D6. This illustrates the utility of obtaining a reference as a result.

Now we’ll show how the INDEX function works with multiple ranges in the reference 
argument. (When you’re using more than one range, you must enclose the argument in 

parentheses.) For example, in the formula =INDEX((A1:C5,D6:F10), 1, 1, 2), the reference 
range comprises two areas: A1:C5 and D6:F10. The area_num argument (2) tells INDEX to 
work on the second of these areas. This formula returns the address D6, which is the cell in 
the first column and first row of the range D6:F10. The displayed result is the value in that 
cell.

The INDIRECT function

The INDIRECT function returns the contents of a cell by using its reference. It takes the 
arguments (ref_text, a1), where ref_text is an A1-style or R1C1-style reference or a cell name. 
The a1 argument is a logical value indicating which type of reference you’re using. If a1 
is FALSE, Excel interprets ref_text as R1C1 format; if a1 is TRUE or omitted, Excel interprets 
ref_text as A1 format. For example, if cell C6 on your worksheet contains the text value B3 
and cell B3 contains the value 2.888, the formula =INDIRECT(C6) returns the value 2.888. 
If your worksheet is set to display R1C1-style references and cell R6C3 contains the text 
reference R3C2 and cell R3C2 contains the value 2.888, then the formula =INDIRECT(R6C3, 
FALSE) also returns the value 2.888.

For information about A1-style and R1C1-style references, see “Understanding the row- 
column reference style” in Chapter 12 .

The ROW and COLUMN functions

The result of the ROW and COLUMN functions is the row or column number, respectively, 
of the cell or range referred to by the function’s single argument. For example, the formula 
=ROW(H5) returns the result 5. The formula =COLUMN(C5) returns the result 3 because 
column C is the third column on the worksheet.

If you omit the argument, the result is the row or column number of the cell that contains 
the function. If the argument is a range or a range name and you enter the function as an 
array by pressing Ctrl+Shift+Enter, the result of the function is an array that consists of the 
row or column numbers of each row or column in the range. For example, suppose you 
select cells B1:B10, type the formula =ROW(A1:A10), and then press Ctrl+Shift+Enter to 
enter the formula in all cells in the range B1:B10. That range will contain the array result 
{1;2;3;4;5;6;7;8;9;10}, the row numbers of each cell in the argument.

The ROWS and COLUMNS functions

The ROWS and COLUMNS functions return the number of rows or columns, respectively, 
referenced by the function’s single argument in a reference or an array. The argument is an 
array constant, a range reference, or a range name. For example, the result of the formula 
=ROWS({100,200,300;1000,2000,3000}) is 2 because the array consists of two rows (sepa-
rated by a semicolon). The formula =ROWS(A1:A10) returns 10 because the range A1:A10 

contains ten rows. And the formula =COLUMNS(A1:C10) returns 3 because the range 
A1:C10 contains three columns.

The AREAS function

You can use the AREAS function to determine the number of areas in a reference. Areas 
refer to individual cell or range references, not regions. The single argument to this func-
tion can be a cell reference, a range reference, or several range references. If you use sev-
eral range references, you must enclose them in a set of parentheses so that Excel doesn’t 
misinterpret the commas that separate the ranges. (Although this function takes only one 
argument, Excel still interprets unenclosed commas as argument separators.) For example, 
suppose you assign the name Test to the group of ranges A1:C5,D6,E7:G10. The function 
=AREAS(Test) returns 3, the number of areas in the group.

The TRANSPOSE function

The TRANSPOSE function changes the horizontal or vertical orientation of an array. It takes 
a single argument, array. If the argument refers to a vertically oriented range, the resulting 
array is horizontal. If the range is horizontal, the resulting array is vertical. The first row of 
a horizontal array becomes the first column of the vertical array result, and vice versa. You 
must type the TRANSPOSE function as an array formula in a range that has the same num-
ber of rows and columns as the array argument has columns and rows, respectively.

Note
For quick and easy transposition, select the range you want to transpose, press Ctrl+C 
to copy the range, click the cell where you want the upper-left corner of the transposed 
range to begin, click the Paste button on the Home tab, and then click the Transpose 
button .

CHAPTER 15

Formatting and calculating 
date and time

You Can use date and time values to stamp documents and to perform date and time 
arithmetic. Creating a production schedule or a monthly billing system is relatively 
easy with Microsoft Excel. Although Excel uses numeric values to count each nano-

second, starting from the beginning of the twentieth century, you can use formatting to 
display those numbers in whatever form you want.

Understanding how Excel records dates and times
Excel assigns serial values to days, hours, minutes, and seconds, which makes it possible for 
you to perform sophisticated date and time arithmetic. The basic unit of time in Excel is 
the day. Each day is represented by a serial date value. The base date, represented by the 
serial value 1, is Sunday, January 1, 1900. When you enter a date in your worksheet, Excel 
records the date as a serial value that represents the number of days between the base date 
and the specified date. For example, Excel represents the date January 1, 2014, by the serial 
value 41640, representing the number of days between the base date—January 1, 1900—
and January 1, 2014.

The time of day is a decimal value that represents the portion of a day that has passed since 
the day began—midnight—to the specified time. Therefore, Excel represents noon by the 
value 0.5 because the difference between midnight and noon is exactly half a day. Excel 
represents the time/date combination 12:59:54 PM, January 1, 2014, by the serial value 
41640.54159 because January 1, 2014, is day 41640 (counting January 1, 1900, as day 1), 
and the interval between midnight and 12:59:54 PM amounts to .54159 of a whole day.

Note
You can see the serial value of a formatted date by selecting the cell containing the 
date and pressing Ctrl+Shift+tilde (~) . Or click the Number Format drop-down list on 
the Home tab and select the General format . Either way, to return the cell to its date 
format, press Ctrl+Z .

